Demonstration of the effect of flow regime on pressure drop

This notebook has been written in *Mathematica* by

Mark J. McCready
Professor and Chair of Chemical Engineering
University of Notre Dame
Notre Dame, IN 46556
USA

mjm@nd.edu
http://www.nd.edu/~mjm/

Version: 12/28/98

Summary

This notebook is intended to give a first introduction to multifluid flows through the use of "model" *flow regimes* calculated from exact solutions for laminar flow in different configurations. By comparing pressure drop over a range of flow rates for these different configurations, that **show differences of factors of up to 30**, the importance of knowing the flow regime is demonstrated. Insight into the physical reasons for the variation in pressure drop with flow rates and physical properties is given.

How to use this notebook

The best way to use this notebook is to open it in *Mathematica* and work through the examples and make changes in parameters or procedural steps to *explore* these problems. On line help is available in *Mathematica* 3 so that a definition and in most cases examples for any unknown command can be obtained. If you would like to view the movies of flow regimes you will need web accesss and a web browser (which you might need to have turned on first). If you do not have a license for *Mathematica*, you can download MathReader (http://www.wolfram.com/mathreader/) free of charge. It does not let you change anything or run the calculations, but it does allow full access to the notebook.

If you have to, you might be able to get some of the message from reading the .pdf file.

Other notebooks that cover a range of fluid dynamics problems are available at:
http://www.nd.edu/~mjm/

These can also be used to explore multifluid flows and learn more about using *Mathematica*.

Motivation: Importance of multifluid flows

There are several reasons that multifluid flows are of interest. In the process industries, operations such as extraction, distillation and stripping involve contacting of immiscible fluids and thus multifluid flows. Key questions are the amount of interfacial area produced by various mixing schemes, the interfacial transport rates and possibly pressure drop for pumping. In the petroleum industry, long distance transportation by pipe of gas-liquid mixtures is quite common. Prediction of flow regime and pressure drop — flowrate relations are crucial to success of design and operation -- particularly of facilities to produce oil from deep water.

The academic interest of multifluid flows stems from the inherent complexity of these flows. For a horizontal two-fluid flow in a simple geometry (circular or 2-D), there are six dimensionless groups. Thus a general predictive relation for say, flowrate — pressure drop, cannot be obtained in a simple a form such as the f - Re relation for single phase flow. With two or more fluids, the new problems that arise and really make the problem more interesting (and hard) are the presence of *interface* with dynamics different from either single phase and the issue of the location of the phases.

Importance of knowing the flow regime

Prediction of flow regime for process and pipeline flows is still considered one of the major unsolved problems of multifluid flows because prediction procedures typically have limited (and sometimes unknown) ranges of validity and the regime exerts a zero order effect on the important macroscopic properties of the flow.

Examples of flow regimes (video clips)

[BACK] to "How to"

When a gas and a liquid are forced to flow together inside a pipe, there are at least 7 different geometrical configurations, or flow regimes, that are observed to occur. The regime depends on the fluid properties, the size of the conduit and the flow rates of each of the phases. The flow regime can also depend on the configuration of the inlet; the flow regime may take some distance to develop and it can change with distance as (perhaps) the pressure, which affects the gas density, changes. For fixed fluid properties and conduit, the flow rates are the independent variables that when adjusted will often lead to changes in the flow regime.

Air-water flow in a 1.27 cm diameter pipe oriented horizontally.

These videos are from **"Two Phase Flow Regimes in Reduced Gravity", NASA Lewis Research Center Motion Picture Directory 1704**.

By **J. B. McQuillen, R. Vernon and A. E. Dukler.**

The video was taken at 400 fps and the projection is at 29.97 fps

Earth gravity

Bubbly flow(.mov movie)

Superficial gas velocity = .16 m/s, Superficial liquid velocity = .90 m/s.

In this example of bubbly flow, the liquid flow rate is high enough to break up the gas into bubbles, but it is not high enough to cause the bubbles to become mixed well within the liquid phase. The Froude number,

$$F = \frac{g\,d}{{U_L}^2} = .15$$

where g is gravitation acceleration, d is the pipe diameter and U_L is liquid superficial velocity, shows that while inertia is slightly larger than gravity forces, the experiments show that there is not enough mixing to scatter the bubbles.

If the pipe were oriented vertically, the phase orientation would be symmetric, but there would likely be "slip" between the phases and the gas would not move at the same speed as the liquid.

Annular Flow (.mov movie)

Superficial gas velocity = 7.4 m/s, Superficial liquid velocity = .08 m/s.

In annular flow, the liquid coats the walls. However, because of gravity, the liquid distribution is not symmetric. There is much more liquid on the bottom of the pipe than the top. The velocity of the gas is large enough to cause waves to form in the liquid and also to atomize some liquid. The maximum possible wave amplitudes scale, for liquid layers that are not too thin, as roughly the liquid thickness.

Slug flow (.mov movie)

Superficial gas velocity = .17 m/s, Superficial liquid velocity = .08 m/s.

The slug regime, is characterized by the presence of liquid rich *slugs* that span the entire channel or pipe diameter. These travel at a speed that is a substantial fraction of the gas velocity and they occur intermittently. Slugs cause large pressure and liquid flow rate fluctuations. The movie shows the approach of first a large wave and later a long slug. Other movies of slugs would show much more gas entrainment and a flow that looks much more violent. The length to diameter ratio of slugs varies greatly with flow rates, pipe diameter and fluid properties. If the diameter is very large, F can always be large and slug flow, where the entire diameter is bridged, will not form. Instead roll waves waves, which are breaking traveling waves, will be seen. Liquid may or may not coat the entire pipe because there will be substantial atomization.

Here are the same flow rates if gravity is reduced to an insignificant level.

Bubbly flow

Superficial gas velocity = .13 m/s, Superficial liquid velocity = .89 m/s.

In this example of bubbly flow, there is no gravity so that there is no buoyancy force on the gas bubbles. Thus they mix freely within the liquid. While there is definitely a continuous phase (liquid) and a dispersed phase (gas), this is close to the idealized *homogenous* or *dispersed* flow that will be used for comparison in the examples below.

Annular flow

Superficial gas velocity = 8.1 m/s, Superficial liquid velocity = .08 m/s.

Now that gravity is removed, the liquid distribution is uniform around the pipe. Large disturbance waves still occur but they are seen as "ring - like" disturbances. The absence of gravity also increases the amplitude to film depth ratio of traveling waves because there is no liquid drainage from the wave caused by gravity.

Slug Flow

Superficial gas velocity = .16 m/s, Superficial liquid velocity = .08 m/s.

In the absence of gravity, the liquid distribution is uniform and the slugs are now liquid "trapped" between traveling "Taylor Bubbles". This flow will not experience large pressure fluctuations and the flow rate fluctuations occur only on the size of the bubbles. This is close to the idealized *slug* regime that is considered in the calculations below.

The calculations below show directly that for the easily calculable case of laminar flow, the flow regime greatly influences the pressure - drop flow rate relation. Thus, if design or operation of a device requires accurate knowledge of flow rate and pressure drop, there is a need to know the flow regime. The rates of heat and mass transfer are also often important in process equipment and the movies suggest that these will also depend significantly on the flow regime!

Intent of this notebook

This notebook is intended to serve as a basic introduction into one aspect of multifluid flows – the importance of knowing the flow geometry, that is the *flow regime,* when a flowrate – pressure drop relation is needed. To do this we use three idealized flows that have (more or less) exact solutions of the governing equations. It should be noted that for various parameter ranges it may not be possible to physically construct the flows that are being computed. However, it should also be realized that this demonstration exercise, **does not** in any way, **overemphasize** the importance of correct knowledge that the *regime* has on the design of a process flow!

There are very few exact solutions of the governing equations for multifluid flows. Here we use one exact solution, a two-layer laminar, *stratified* flow and two idealized solutions, an alternating fluid1-fluid2 laminar flow (which we term a *slug* flow) and a *dispersed* fluid1-fluid 2 flow that is laminar with fluid properties described by an average viscosity and density.

Our goal is to first show the extent of the predicted difference in the pressure drop depending upon the arrangement of the phases. Second we would like to gain some physical understanding of the differences between multifluid flows and single phase flows. Finally, some effort is given to demonstrating various mathematical manipulations that are useful for studying the equations of multifluid flows.

Preview of Major points

We have shown, using simple models for flow regimes, stratified, slug and dispersed, that

1. The *qualitative* as well as the *quantitative* behavior of multiphase flows will change as the ratios of flow rates and physical properties change.

2. The pressure drop predictions differ substantially with flow configuration. The pressure drop for dispersed flow was predicted to be a factor of 35 higher than for slug flow in one case and a factor of 20 greater than stratified flow for another case This key result is true for process flows and makes correct prediction of the flow regime crucial to successful design of multifluid systems. Most engineering designs cannot stand an uncertainty of a factor of 2 in the main design variable, let alone 30.

3. Stratified flow is the most efficient configuration, of the three tested here (compare stratified/slug, dispersed/stratified), for fluid transport when the more viscous fluid has a higher flow rate. This is due to the lubricating effect of the less viscous fluid that reduces shear in the more viscous fluid. This is the basis of lubricated pipeline transport of heavy oil (See D. D. Joseph and Y. Y. Renardy, *Fundamentals of Two-Fluid Dynamics*, Springer-Verlag, 1993, Vol. 2.) If the more viscous fluid is present in lessor amounts the advantage is lost because it is subjected to high shear and acts to reduce the available flow area for the less viscous fluid.

4. The loss of lubricating effect of a less viscous fluid in stratified flow can cause a region where *decreasing* the flow rate of the less viscous fluid, *increases* the pressure drop (click for specifics about *retrograde* pressure drop) -- contrary to physical intuation gained from most other flow situations.

5. The specific conclusions for dispersed/slug, dispersed/stratifed and stratified/slug can be accessed directly.

6. The reason for the differences in the pressure drop with configuration for the examples in this notebook is that the dissipation is altered. Differences in dissipation arise primarily when fluids of different viscosities are located in regions of different stress. We also find that changing the effective flow area (i.e., by having a stratified region of more viscous fluid) for the fast moving fluid changes the dissipation significantly. These general observations should hold for either laminar flow (as shown here) or turbulent flow. However, if the primary contributions to pressure drop are from fluid acceleration, or gravity, then the pressure drop differences caused by the flow regime could be less than shown here. Examples are unsteady or transient flows, developing flows or vertical flows.

[BACK to Recap]

Flow rate pressure drop relations for the three regimes.

Stratified flow



[BACK to Preview]
[BACK to Recap]

Problem description

As shown in the picture, we consider a two-layer, steady, laminar flow with a flat interface. The two fluids are presumed to have different viscosities and densities (and be immiscible) so that the separate regions are governed by separate flow equations. The no-slip condition will be enforced at the lower ($y=-h$) and upper ($y = H-h$) walls. At the interface ($y = 0$) we will have no slip between the fluids and a continuity of shear stress. For two, 2nd order ODE's, this gives us the required number of boundary conditions.

Experimental and process flows are often stratified although the usual case would be that waves form. (Consider also the case of wind blowing over a body of water. There will be a current caused by the wind and waves may form.) If you would like to see still pictures of waves or video clips of waves, please click on the highlighted text.

Mathematica aside

In *Mathematica*, it is convenient to give all expressions a "name". I try to pick ones that are consistent with what is being done (but sometimes "temp" is used). This assignment is done with an "=" sign. To make an equation, a "==" is used. This distinction is very useful in computer algebra and is employed in all of the packages with which I am familiar .

Equations and boundary conditions

Governing equations for each fluid

```
strateq1 = μL ∂{y,2} uL[y] - dpdxL
```

$\mu_L\, u_L''(y) - \mathrm{dpdx}_L$

```
strateq2 = μG ∂{y,2} uG[y] - dpdxG
```

$\mu_G\, u_G''(y) - \mathrm{dpdx}_G$

Boundary conditions

Bottom wall

```
bc1 = uL[-h] == 0
```

$u_L(-h) == 0$

top wall

```
bc2 = uG[H - h] == 0
```

$u_G(H - h) == 0$

- **The velocity and stress match at the interface.**

```
bc3 = uL[0] == uG[0]
```

$u_L(0) == u_G(0)$

```
bc4 = (μL D[uL[y], y] /. y -> 0) == (μG D[uG[y], y] /. y -> 0)
```

$\mu_L\, u_L'(0) == \mu_G\, u_G'(0)$

- **Velocity profiles**

We can solve all of the above equations together and simplify the result by using the command:

```
stratans1 =
 Simplify[DSolve[{strateq1 == 0, strateq2 == 0, bc1, bc2, bc3, bc4},
 {uL[y], uG[y]}, y]]
```

$$\left\{\left\{u_L(y) \to \frac{(h+y)\left((h-H)^2\, \text{dpdx}_G\, \mu_L - \text{dpdx}_L\, (h\, y\, \mu_G + (h-H)\,(h-y)\, \mu_L)\right)}{2\, \mu_L\, ((h-H)\, \mu_L - h\, \mu_G)},\right.\right.$$
$$\left.\left. u_G(y) \to \frac{(h-H+y)\left(\text{dpdx}_L\, \mu_G\, h^2 + \text{dpdx}_G\, (h\,(-h+H+y)\, \mu_G + (H-h)\, y\, \mu_L)\right)}{2\, \mu_G\, (h\, \mu_G + (H-h)\, \mu_L)}\right\}\right\}$$

We should check the profiles to see if they make sense. Here we use an oil-water flow in a 1 cm channel.

```
testplot1 = Plot[
  (uL[y] /. stratans1[[1]]) /. {H -> 1, h -> .4, dpdxL -> -.2, μG -> .2,
   dpdxG -> -.2, μL -> .01}, {y, -.4, 0}]
```



- Graphics -

```
testplot2 = Plot[
  (uG[y] /. stratans1[[1]]) /. {H -> 1, h -> .4, dpdxL -> -.2, μG -> .2,
   dpdxG -> -.2, μL -> .01}, {y, 0, .6}]
```



- Graphics -

We see nice profiles for this oil-water flow with water on the left (the lower region).

```
Show[testplot1, testplot2]
```



- Graphics -

- **Average velocities**

We need the average velocities to make comparisons based on flow rates.

```
ulstratave = Simplify[∫_{-h}^{0} (u_L[y] /. stratans1[[1]]) dy / h]
```

$$\frac{h\,(3\,\mathrm{dpdx}_G\,\mu_L\,(h-H)^2 + h\,\mathrm{dpdx}_L\,(h\,\mu_G + 4\,(H-h)\,\mu_L))}{12\,\mu_L\,((h-H)\,\mu_L - h\,\mu_G)}$$

```
ugstratave =
 Simplify[Integrate[(u_G[y] /. stratans1[[1]]), {y, 0, H - h}] / (H - h)]
```

— *General::spell1 : Possible spelling error: new symbol name "ugstratave" is similar to existing symbol "ulstratave".*

$$-\frac{(h-H)\,((h-H)\,\mathrm{dpdx}_G\,(4\,h\,\mu_G + (H-h)\,\mu_L) - 3\,h^2\,\mathrm{dpdx}_L\,\mu_G)}{12\,\mu_G\,(h\,\mu_G + (H-h)\,\mu_L)}$$

- **Pressure drop in terms of flow rates**

In an experimental or process flow, the flow rates for the fluids are chosen and the pressure drop and depths of the fluids adjust to appropriate values. Thus we would like to choose the liquid and gas flow rates as input variables and eliminate *h* and then solve for dpdx. Unfortunately, we cannot do this analytically as it would produce a 7th order equation for *h*. Thus, let us eliminate dpdx and then solve for U_G. We can later translate as needed to compare to Re_L, Re_G input info.

We realize that if the flow is horizontal, there is no hydraulic gradient and thus $dpdx_L = dpdx_G$.

Start with the average equation for the lower fluid,

```
dpeq1 = (ulstratave /. {dpdxG -> dpdx, dpdxL -> dpdx}) == ŪL
```

$$\frac{h\,(3\,\text{dpdx}\,\mu_L\,(h-H)^2+\text{dpdx}\,h\,(h\,\mu_G+4\,(H-h)\,\mu_L))}{12\,\mu_L\,((h-H)\,\mu_L-h\,\mu_G)} == \overline{U}_L$$

Solve this for the pressure drop. Note that if *h* and $\bar{U}_L$ are known, the flow is completely prescribed.

```
presstemp1 = Solve[dpeq1, dpdx]
```

$$\left\{\left\{\text{dpdx}\to-\frac{12\,\mu_L\,(-h\,\mu_G+h\,\mu_L-H\,\mu_L)\,\overline{U}_L}{h\,(-\mu_G\,h^2+\mu_L\,h^2+2\,H\,\mu_L\,h-3\,H^2\,\mu_L)}\right\}\right\}$$

Then we get the stratified pressure drop in a useful form for later calculations.

```
dpdxstrat = Simplify[(dpdx /. presstemp1[[1]]) /. ŪL -> ReL μL / h / ρL]
```

$$-\frac{12\,\text{Re}_L\,\mu_L^2\,((h-H)\,\mu_L-h\,\mu_G)}{h^2\,((h^2+2\,H\,h-3\,H^2)\,\mu_L-h^2\,\mu_G)\,\rho_L}$$

Use the pressure drop in the equation for the average velocity of the second fluid to get a useful relation for its average velocity.

```
Ugexpress = Simplify[((ugstratave /. {dpdxG -> dpdx, dpdxL -> dpdx}) /.
     presstemp1[[1]]) /.
    ŪL -> ReL μL / h / ρL]
```

$$-\frac{(h-H)\,\mathrm{Re}_L\,\mu_L^2\,((h-H)^2\,\mu_L-h\,(h-4\,H)\,\mu_G)}{h^2\,\mu_G\,(h^2\,\mu_G-(h^2+2\,H\,h-3\,H^2)\,\mu_L)\,\rho_L}$$

■ **Friction velocity**

It is useful to also calculate the interfacial friction velocity. It is often termed, v*, and is defined as $\sqrt{\frac{\tau}{\rho}}$ for τ the interfacial shear and ρ the liquid density.

```
vstarL =
 Simplify[((Sqrt[μL (D[uL[y] /. stratans1[[1]], y] /. y -> 0) / ρL] /.
      {dpdxG -> dpdx, dpdxL -> dpdx}) ) /.
   presstemp1[[1]]]
```

$$\sqrt{6}\,\sqrt{-\frac{\mu_L\,((h-H)^2\,\mu_L-h^2\,\mu_G)\,\overline{U}_L}{h\,((h^2+2\,H\,h-3\,H^2)\,\mu_L-h^2\,\mu_G)\,\rho_L}}$$

Dispersed



[Back to Preview]
[BACK to Recap]

View the bubbly flow movie, Bubbly flow

Problem description

We choose a channel is *H* high with fixed wall boundary conditions and the origin on the bottom wall. We presume that we can describe a dispersed flow with an average viscosity and density. Note that this works sometimes, but in real flows there are several problems. First, it should be recognized, as mentioned above, that it may not be possible to construct a uniformly dispersed (homogeneous) flow at all flow rates. Second even if the flow exists, there can be an effective "slip" between the phases so that a "drift flux" model is needed (see *Two-phase flows and heat transfer with applications to nuclear reactor design problems*, ed. J. J. Ginoux, Hemisphere, 1978, pp35-43). Third even if there is little slip between the phases, the average viscosity of the mixture can be a complicated and often unknown function (ibid p22; M. Ishii & N. Zuber *AIChE J*. **25** pp843-854, 1979). Finally, even when the flow is nominally dispersed, it may not be homogeneous so that people solve averaged versions of the Navier-Stokes equations for each phase with appropriate closures.

Equations for a steady, laminar single phase flow

```
dispeq1 = μmix ∂{y,2} umix[y] - dpdxmix
```

$$\mu_{\text{mix}}\, u''_{\text{mix}}(y) - \text{dpdx}_{\text{mix}}$$

The single differential equation and no slip boundary conditions are easily solved.

```
ans1 = DSolve[{dispeq1 == 0, umix[0] == 0, umix[H] == 0}, umix[y], y]
```

$$\left\{\left\{u_{\text{mix}}(y) \to \frac{y^2\, \text{dpdx}_{\text{mix}}}{2\, \mu_{\text{mix}}} - \frac{H\, y\, \text{dpdx}_{\text{mix}}}{2\, \mu_{\text{mix}}}\right\}\right\}$$

We like to extract the velocity as

```
udisp = umix[y] /. ans1[[1]]
```

$$\frac{y^2\, \text{dpdx}_{\text{mix}}}{2\, \mu_{\text{mix}}} - \frac{H\, y\, \text{dpdx}_{\text{mix}}}{2\, \mu_{\text{mix}}}$$

Average velocity

We need the average velocity to get the flow rate.

```
umixave = ∫_0^H udisp ⅆy / H
```

$$-\frac{H^2\,\mathrm{dpdx}_{\mathrm{mix}}}{12\,\mu_{\mathrm{mix}}}$$

- **Pressure drop**

We can rearrange to get the pressure drop, and give it the new name, $\mathrm{dpdx}_{\mathrm{disp}}$.

Note the the command below says to make an equation with the expression for umixave, with $\bar{U}_{\mathrm{mix}}$, solve it for $\mathrm{dpdx}_{\mathrm{mix}}$, then take the first part of it, which is the piece inside the outside braces, and use this as a substitution rule to replace $\mathrm{dpdx}_{\mathrm{mix}}$. This expression has the new name, $\mathrm{dpdx}_{\mathrm{disp}}$ which will be used below.

```
dpdx_disp = dpdx_mix /. Solve[umixave == Ū_mix, dpdx_mix][[1]]
```

— *General::spell1 : Possible spelling error: new symbol name "disp" is similar to existing symbol "udisp".*

$$-\frac{12\,\mu_{\mathrm{mix}}\,\overline{U}_{\mathrm{mix}}}{H^2}$$

- **Friction factor**

We might be interested in the friction factor, which is defined as

```
f_mix = -H dpdx_disp / 2 / ρ_mix / Ū_mix^2
```

$$\frac{6\,\mu_{\mathrm{mix}}}{H\,\rho_{\mathrm{mix}}\,\overline{U}_{\mathrm{mix}}}$$

Which is the expected $f = 6/Re$ for a channel flow.

- **Friction velocity**

It could also be of interest to derive the wall friction velocity.

```
vstarmix = Sqrt[μmix (D[udisp, y] /. y -> 0) / ρmix] /. dpdxmix -> dpdxdisp
```

$$\sqrt{6}\sqrt{\frac{\mu_{mix}\,\overline{U}_{mix}}{H\,\rho_{mix}}}$$

Slug (Alternating)



View the slug flow movie, Slug Flow

[BACK to Preview]
[BACK to Recap]

Problem description

We cannot easily compute the exact relation for this alternating geometry even for laminar flow, although a number of solutions exist for bubbles in capillaries at low Reynolds number. We will instead restrict our solution to the case where the slugs and bubbles are long compared to the diameter (not like the picture) so that each region is in fully-developed laminar flow. We can then get an average pressure drop using the volumetric flowrate to tell us the fraction of the time that each region is contributing.

Equations for a steady, laminar single phase flow

```
slugeq1 = μ ∂{y,2} u[y] - dpdx
```

$\mu\, u''(y) - \text{dpdx}$

We choose a channel that is H high with fixed wall boundary conditions and the origin on the bottom wall.

```
slugans1 = DSolve[{slugeq1 == 0, u[0] == 0, u[H] == 0}, u[y], y]
```

$$\left\{\left\{u(y) \to \frac{\text{dpdx}\, y^2}{2\,\mu} - \frac{\text{dpdx}\, H\, y}{2\,\mu}\right\}\right\}$$

- **Velocity profile**

```
slugans2 = u[y] /. slugans1[[1]]
```

$$\frac{\text{dpdx}\, y^2}{2\,\mu} - \frac{\text{dpdx}\, H\, y}{2\,\mu}$$

- **Average velocity**

We need the average velocity to get the flow rate.

$$\text{uslugave} = \frac{\int_0^H \text{slugans2}\, dy}{H}$$

$$-\frac{\text{dpdx}\, H^2}{12\,\mu}$$

- **Pressure drop**

Rearrange the average velocity equation to get the pressure drop.

$$\text{dpdx}_{\text{slug}} = \text{dpdx} \;/.\; \text{Solve}[\text{uslugave} == \bar{U}, \text{dpdx}][[1]]$$

$$-\frac{12\,\mu\,\overline{U}}{H^2}$$

- **Friction factor**

It is convenient to construct a friction factor – Reynolds number relation from this.

```
ftemp = -H dpdx_slug / 2 / ρ / Ū^2
```

$$\frac{6\,\mu}{H\,\rho\,\overline{U}}$$

Which is the expected $f = 6/Re$ for a channel flow.

Friction velocity

It is also useful to derive the friction velocity.

```
vstar_slug = Sqrt[μ (D[slugans2, y] /. y -> 0) / ρ] /. dpdx -> dpdx_slug
```

$$\sqrt{6}\,\sqrt{\frac{\mu\,\overline{U}}{H\,\rho}}$$

Pressure drop comparisons

We now need to compare the pressure drop for the different configurations to see what happens. To show the importance of the regime, the ratio of pressure drop between two different regimes will be plotted as a function of flow rate.

Dispersed / Slug

Pressure drop ratio

The pressure drop ratio for dispersed flow to slug flow is given as

```
dpdx_disp / dpdx_slug
```

$$\frac{\mu_{\text{mix}}\,\overline{U}_{\text{mix}}}{\mu\,\overline{U}}$$

Sensible values for the velocities and viscosities are needed in this expression.

Dispersed relations

For the mixture velocity, we just use the total flow rates

$$\text{Umix} = \frac{\frac{\text{Re}_L\ \mu_L}{\rho_L} + \frac{\text{Re}_G\ \mu_G}{\rho_G}}{H}$$

— *General::spell1 : Possible spelling error: new symbol name "Umix" is similar to existing symbol "mix".*

$$\frac{\frac{\text{Re}_G\ \mu_G}{\rho_G} + \frac{\text{Re}_L\ \mu_L}{\rho_L}}{H}$$

Averages will be used for the viscosity and density. A straight volume average should be good for the density. This will not be good for the viscosity and we will use a mass fraction weighting for viscosity.

The volume averaged expression is ξ_{average} = volfrac(L) ξ_L + volfrac(G) ξ_G. The volume fraction for, say, the L phase is $\frac{q_L}{q_L + q_G}$.

The expression for the density is,

$$\rho\text{mix} = \frac{\frac{\text{Re}_L\ \mu_L}{\rho_L}\rho_L}{\left(\frac{\text{Re}_L\ \mu_L}{\rho_L} + \frac{\text{Re}_G\ \mu_G}{\rho_G}\right)} + \frac{\frac{\text{Re}_G\ \mu_G}{\rho_G}\rho_G}{\left(\frac{\text{Re}_L\ \mu_L}{\rho_L} + \frac{\text{Re}_G\ \mu_G}{\rho_G}\right)};$$

For the viscosity we will use an expression (from *Two-phase flows and heat transfer with applications to nuclear reactor design problems*, ed. J. J. Ginoux, Hemisphere, 1978, p22), attributed to Cicchitti et al.. It is essentially a mass fraction weighting and assumes that the more viscous fluid is also more dense. (Of course this is not always true!) If you don't like this one, check out the above-mentioned Ishii and Zuber paper.

$$\mu\text{mix} = \frac{\rho_L \frac{\text{Re}_L\ \mu_L}{\rho_L}}{\left(\rho_L \frac{\text{Re}_L\ \mu_L}{\rho_L} + \rho_G \frac{\text{Re}_G\ \mu_G}{\rho_G}\right)}\mu_L + \frac{\rho_G \frac{\text{Re}_G\ \mu_G}{\rho_G}}{\left(\rho_L \frac{\text{Re}_L\ \mu_L}{\rho_L} + \rho_G \frac{\text{Re}_G\ \mu_G}{\rho_G}\right)}\mu_G;$$

— *General::spell1 : Possible spelling error: new symbol name "μmix" is similar to existing symbol "ρmix".*

```
μmixsimp = Simplify[μmix]
```

$$\frac{\text{Re}_G\ \mu_G^2 + \text{Re}_L\ \mu_L^2}{\text{Re}_G\ \mu_G + \text{Re}_L\ \mu_L}$$

Now substitute the averages into the expression for the dispersed pressure drop.

```
disptemp = FullSimplify[dpdx_disp /. {μ_mix -> μmixsimp,
   Ū_mix -> Umix}]
```

$$-\frac{12\,(\mathrm{Re}_G\,\mu_G^2 + \mathrm{Re}_L\,\mu_L^2)\left(\frac{\mathrm{Re}_G\,\mu_G}{\rho_G} + \frac{\mathrm{Re}_L\,\mu_L}{\rho_L}\right)}{H^3\,(\mathrm{Re}_G\,\mu_G + \mathrm{Re}_L\,\mu_L)}$$

- **Slug relations**

```
dpdx_slug
```

$$-\frac{12\,\mu\,\overline{U}}{H^2}$$

From the relation for $dpdx_{slug}$, we see that we need an average of the product of μ and $\overline{U}$. The way this works is that the pressure drop for each region is just $\frac{-12\,\mu_i\,U_i}{H^2}$ but the U_i is increased over its single phase velocity by the loss of flow area owing to the presence of the other phase. So in a region of L, the pressure drop is higher than if no G were present. Of course, the entire pipe is not filled with L, it is only $\frac{U_L}{(U_L+U_G)}$ full of L. So we can combine this to get the final result.

The fraction still available to fluid L should be

```
XL = 1 - U_G / (U_L + U_G)
```

$$1-\frac{U_G}{U_G+U_L}$$

The fraction available for fluid G is then

```
XG = 1 - U_L / (U_L + U_G)
```

$$1-\frac{U_L}{U_G+U_L}$$

This gives a pressure drop that becomes

$$\text{dpslug} = \frac{-12}{H^2}\left(\frac{\mu_L U_L}{XL}\frac{U_L}{(U_L+U_G)} + \frac{\mu_G U_G}{XG}\frac{U_G}{(U_L+U_G)}\right)$$

$$-\frac{12\left(\frac{\mu_G U_G^2}{(U_G+U_L)\left(1-\frac{U_L}{U_G+U_L}\right)} + \frac{U_L^2 \mu_L}{(U_G+U_L)\left(1-\frac{U_G}{U_G+U_L}\right)}\right)}{H^2}$$

$$\text{slugtemp} = \text{Simplify}\left[\text{dpslug} \;/.\; \left\{U_L \to \frac{\text{Re}_L\, \mu_L}{H\, \rho_L},\; U_G \to \frac{\text{Re}_G\, \mu_G}{H\, \rho_G}\right\}\right]$$

$$-\frac{12\left(\text{Re}_G\, \rho_L\, \mu_G^2 + \text{Re}_L\, \mu_L^2\, \rho_G\right)}{H^3\, \rho_G\, \rho_L}$$

This is, of course, the same as the obvious answer,

$$\text{dpslug2} = \text{FullSimplify}\left[\left(\frac{-12}{H^2}(\mu_L U_L + \mu_G U_G)\right) \;/.\; \left\{U_L \to \frac{\text{Re}_L\, \mu_L}{H\, \rho_L},\; U_G \to \frac{\text{Re}_G\, \mu_G}{H\, \rho_G}\right\}\right]$$

$$-\frac{12\left(\text{Re}_G\, \rho_L\, \mu_G^2 + \text{Re}_L\, \mu_L^2\, \rho_G\right)}{H^3\, \rho_G\, \rho_L}$$

- **The dispersed- slug ratio**

```
dispslugratio = Simplify[disptemp / slugtemp]
```

$$\frac{\left(\text{Re}_G\, \mu_G^2 + \text{Re}_L\, \mu_L^2\right)\left(\text{Re}_L\, \mu_L\, \rho_G + \text{Re}_G\, \mu_G\, \rho_L\right)}{\left(\text{Re}_G\, \mu_G + \text{Re}_L\, \mu_L\right)\left(\text{Re}_G\, \rho_L\, \mu_G^2 + \text{Re}_L\, \mu_L^2\, \rho_G\right)}$$

- **Limits of the expression**

This result gives all of the expected limits.

```
ratiotest = FullSimplify[dispslugratio /. {ReL -> UL ρL H / μL,
   ReG -> UG ρG H / μG}]
```

$$\frac{(U_G + U_L)\,(U_G\,\mu_G\,\rho_G + U_L\,\mu_L\,\rho_L)}{(U_G\,\mu_G + U_L\,\mu_L)\,(U_G\,\rho_G + U_L\,\rho_L)}$$

```
ratiotest2 = FullSimplify[ratiotest /.
   { μG -> μL }]
```

1

```
Limit[dispslugratio, ReG -> 0]
```

1

```
Limit[dispslugratio, ReG -> Infinity]
```

1

- **Plots of the ratio**

Air-water in a 2.54 cm channel, Re_L =100.
The pressure drop ratio for dispersed flow divided by slug flow gives a difference of a factor greater than 30. The liquid flow rate is held constant as the gas flow is increased.

▪ Pressure drop comparison for dispersed vs. slug flow

```
Plot[Evaluate[dispslugratio /. {ReL -> 100, μL -> .01, μG -> .00018,
   ρL -> 1, ρG -> 1 / 899, H -> 1}], {ReG, 1, 10000},
 AxesLabel -> {"ReG", "dispersed/slug ratio"}]
```



- Graphics -

[BACK to Preview]
[BACK to Recap]

Air-water in a 2.54 cm channel, Re_L =1000.

```
Plot[Evaluate[dispslugratio /. {ReL -> 1000, μL -> .01, μG -> .00018,
   ρL -> 1, ρG -> 1 / 899, H -> 1}], {ReG, 1, 10000},
 AxesLabel -> {"ReG", "dispersed/slug ratio"}]
```



- Graphics -

Check to see if anything is happening at small Re_G .

Air-water in a 2.54 cm channel, Re_L =1000.

```
Plot[Evaluate[dispslugratio /. {ReL -> 1000, μL -> .01, μG -> .00018,
   ρL -> 1, ρG -> 1 / 899, H -> 1}], {ReG, .01, 1},
 AxesLabel -> {"ReG", "dispersed/slugratio"}]
```



- Graphics -

We see that for large Re_G, there is a huge disparity in the predictions of pressure drop.

Oil-water in a 1 cm channel, Re_L =100. Now the dispersed flow has a lower pressure drop.

```
Plot[Evaluate[dispslugratio /. {ReL -> 100, μL -> .01, μG -> .2,
    ρL -> 1, ρG -> .88, H -> 1}], {ReG, .1, 10},
 AxesLabel -> {"Reoil", "dispersed/slug ratio"}]
```



- Graphics -

Oil-water in a 1 cm channel, Re_L =1000. Again, dispersed is lower.

```
Plot[Evaluate[dispslugratio /. {ReL -> 1000, μL -> .01, μG -> .2,
    ρL -> 1, ρG -> .88, H -> 1}], {ReG, .1, 10},
 PlotRange -> All, AxesLabel -> {"Reoil", "dispersed/slug ratio"}]
```



- Graphics -

Oil-water in a 1 cm channel, Re_L =100. There is a modest minimum in the pressure drop ratio.

```
Plot[Evaluate[dispslugratio /. {ReL -> 100, μL -> .01, μG -> .2,
   ρL -> 1, ρG -> .88, H -> 1}], {ReG, .1, 5},
 PlotRange -> All, AxesLabel -> {"Reoil", "dispersed/slug ratio"}]
```



- Graphics -

Dispersed - slug observations

The first observation, that was probably obvious, is that the character of the flow can change with the relative values of the two flow rates.

For gas-liquid cases, we see that dispersed flow always has a larger pressure drop that slug. This is evidently because the viscosity of the mixture is always higher than the less viscous fluid and thus the pressure drop is always increased over a single phase flow of the less viscous fluid. In contrast, the alternating flow has regions of the low viscosity fluid in laminar flow which gives the lowest possible pressure drop for a requisite fraction of the flow distance.

For oil-water flows, dispersed flow has a lower pressure drop. This is evidently because of the density weighting of the mixture viscosity function which keeps the viscosity below its volume fraction value.

For all of the test cases, there is an intermediate maximum and an intermediate minimum that appear to depend on the viscosity ratio. Let's see if we can make some sense of this.

Analysis

The Reynolds number is not an intrinsically important parameter in this problem so let us do some rearranging.

```
dpmax1 = dispslugratio /. {ReL -> UL ρL H / μL,
   ReG -> UG ρG H / μG}
```

$$\frac{(H\,U_G\,\mu_G\,\rho_G + H\,U_L\,\mu_L\,\rho_L)\,(H\,U_G\,\rho_G\,\rho_L + H\,U_L\,\rho_G\,\rho_L)}{(H\,U_G\,\rho_G + H\,U_L\,\rho_L)\,(H\,U_G\,\mu_G\,\rho_G\,\rho_L + H\,U_L\,\mu_L\,\rho_G\,\rho_L)}$$

Now get the pressure drop ratio completely in terms of parameter ratios. Note we could have done the whole problem this way by using the appropriate nondimensionalization at the beginning.

```
dpmax2 = FullSimplify[dpmax1 /. {μG -> m μL, UG -> ψ UL,
    ρG -> r ρL}]
```

$$\frac{(\psi + 1)\,(m\,r\,\psi + 1)}{(m\,\psi + 1)\,(r\,\psi + 1)}$$

If the viscosities are equal, m=1, then the models give the same result as they should.

```
Simplify[dpmax2 /. m -> 1]
```

1

Likewise, if the densities of the fluids are equal, r=1, the pressure drop ratio is unity.

```
Simplify[dpmax2 /. r -> 1]
```

1

For r not equal to unity, we see that we have the same behavior as above using this simpler expression and we now find that there is a maximum. Note that in the limit of gas/liquid flowrate ratio, $\psi \to \infty$, the pressure ratio returns to unity.

We need to load a package to do a log plot

```
<< Graphics`Graphics`
```

Here is air-water.

```
LogLogPlot[dpmax2 /. {m -> .02, r -> .001}, {ψ, .01, 100000},
 AxesLabel -> {"flow ratio", "dispersed/slug ratio"}]
```



- Graphics -

Here is oil-water

```
LogLinearPlot[dpmax2 /. {m -> 20, r -> .8},
 {ψ, .01, 200}, PlotPoints -> 100, PlotRange -> All,
 AxesLabel -> {"flow ratio", "dispersed/slug ratio"}]
```



- Graphics -

For r<1 and m<1, there is a maximum at high flow rates. For r<1, m>1, there is a minimum.

Now take the derivative to find the extremum,

```
dpmax3 = FullSimplify[D[dpmax2, ψ ]]
```

$$-\frac{(m-1)(r-1)(m\, r\, \psi^2-1)}{(m\, \psi+1)^2 (r\, \psi+1)^2}$$

We are fortunate that we get a simple result for the flow rate ratio, ψ of the extreme point,

```
dpmax4 = Solve[dpmax3 == 0, ψ]
```

$$\left\{\left\{\psi \to -\frac{1}{\sqrt{m}\ \sqrt{r}}\right\}, \left\{\psi \to \frac{1}{\sqrt{m}\ \sqrt{r}}\right\}\right\}$$

We see that ψ, the flow rate ratio, for the maximum pressure drop will be large for air-water flows and possibly closer to unity for oil-water flows for the minimum. The pressure drop value of the maximum or minimum is given by substituting this into the pressure drop ratio expression,

```
maxeq1 = FullSimplify[dpmax2 /. dpmax4[[2]]]
```

$$\frac{\left(\sqrt{m}\ \sqrt{r}+1\right)^2}{\left(\sqrt{m}+\sqrt{r}\right)^2}$$

It is seen that this value can be arbitrarily large, or somewhat smaller than 1. Most engineering designs cannot stand a factor of 20 uncertainty in a key parameter.

Consider gas-liquid density ratio. The air-water viscosity ratio would be 0.018, giving a factor of 30 of difference in the pressure prediction of the two models.

```
LogLogPlot[maxeq1 /. r -> 1 / 900,
 {m, .001, .1}, AxesLabel -> {"viscosity ratio",
   "max dpratio"}]
```



- Graphics -

The minimum for oil-water is much more modest.

```
LogLogPlot[maxeq1 /. r -> .8, {m, 1, 1000},
 PlotRange -> {.3, 1}, AxesLabel -> {"viscosity ratio",
   "min dpratio"}]
```



- Graphics -

- **Conclusions for dispersed verus slug flows**

The dispersed flow has a higher pressure drop than the model slug flow if the less viscous fluid has a lower density than the more viscous fluid. This difference can easily be a factor of 30 for reasonable values of the viscosity and velocity ratios. Dispersed flow gives a modest reduction in pressure drop (~10-15%) occurs (compared to slug) if the more viscous fluid has a lower density. It should be emphasized that the model for the mixture viscosity of the dispersed flow can significantly change these results (Check this yourself).

[BACK to Preview]
[BACK to Recap]

We can examine the viscosity relation for dispersed flow to gain some physical insight into why the dispersed flow can have a much higher pressure drop for air-water.

Change the mixture viscosity into ratios,

```
visc1 = μmix /. {Re_L -> U_L ρ_L H / μ_L,
   Re_G -> U_G ρ_G H / μ_G}
```

$$\frac{H\,U_G\,\mu_G\,\rho_G}{H\,U_G\,\rho_G + H\,U_L\,\rho_L} + \frac{H\,U_L\,\mu_L\,\rho_L}{H\,U_G\,\rho_G + H\,U_L\,\rho_L}$$

Now compare the mixture to the liquid viscosity,

```
visc2 = FullSimplify[(visc1 /. {μ_G -> m μ_L, U_G -> ψ U_L,
    ρ_G -> r ρ_L}) / μ_L]
```

$$\frac{m\,r\,\psi + 1}{r\,\psi + 1}$$

```
LogLinearPlot[visc2 /. {r -> 1 / 900, m -> .02}, {ψ, .01, 100},
 AxesLabel -> {"Air-water ratio", "mixture viscosity"}]
```



- Graphics -

We see that the mixture viscosity is staying close to the liquid value even when the flow has a high volume fraction of the gas. Thus while the alternating flow has regions of low pressure drop, (the gas), the dispersed flow never gets this benefit.

In contrast for oil-water, since water is less viscous and more dense, the mixture viscosity does not increase too quickly as the oil flow is increased so there is a range where the dispersed pressure drop is not as large as the alternating flow.

```
LogLogPlot[
 visc2 /. {r -> .8, m -> 20}, {ψ, .01, 100}, PlotRange -> All,
 AxesLabel -> {"Oil-water ratio", "mixture viscosity"}]
```



- Graphics -

Dispersed / Stratified

Here is the ratio of dispersed flow to stratified flow

```
dpdx_disp / dpdx_strat
```

$$\frac{h^2\,((h^2 + 2\,H\,h - 3\,H^2)\,\mu_L - h^2\,\mu_G)\,\mu_{\text{mix}}\,\rho_L\,\overline{U}_{\text{mix}}}{H^2\,\text{Re}_L\,\mu_L^2\,((h - H)\,\mu_L - h\,\mu_G)}$$

We now need sensible values for the velocities and viscosities.

Dispersed relations

For the mixture velocity, we just use the total flow rates

$$\text{Umix} = \frac{\frac{\text{Re}_L\ \mu_L}{\rho_L} + \frac{\text{Re}_G\ \mu_G}{\rho_G}}{H};$$

Averages will be used for the viscosity and density. A straight volume average should be good for the density. This will not be good for the viscosity and we will use a mass fraction weighting for viscosity.

The volume averaged expression is ξ_{average} = volfrac(L) ξ_L + volfrac(G) ξ_G. The volume fraction for, say, the L phase is $\frac{q_L}{q_L + q_G}$.

The expression for the density is,

$$\rho\text{mix} = \frac{\frac{\text{Re}_L\ \mu_L}{\rho_L}\ \rho_L}{\left(\frac{\text{Re}_L\ \mu_L}{\rho_L} + \frac{\text{Re}_G\ \mu_G}{\rho_G}\right)} + \frac{\frac{\text{Re}_G\ \mu_G}{\rho_G}\ \rho_G}{\left(\frac{\text{Re}_L\ \mu_L}{\rho_L} + \frac{\text{Re}_G\ \mu_G}{\rho_G}\right)};$$

For the viscosity we will use an expression (from *Two-phase flows and heat transfer with applications to nuclear reactor design problems*, ed. J. J. Ginoux, Hemisphere, 1978, p22), attributed to Cicchitti et al.. It is essentially a mass fraction weighting and assumes that the more viscous fluid is also more dense. (Of course this is not always true!) If you don't like this one, check out the above-mentioned Ishii and Zuber paper.

$$\mu\text{mix} = \frac{\rho_L\ \frac{\text{Re}_L\ \mu_L}{\rho_L}}{\left(\rho_L\ \frac{\text{Re}_L\ \mu_L}{\rho_L} + \rho_G\ \frac{\text{Re}_G\ \mu_G}{\rho_G}\right)}\ \mu_L + \frac{\rho_G\ \frac{\text{Re}_G\ \mu_G}{\rho_G}}{\left(\rho_L\ \frac{\text{Re}_L\ \mu_L}{\rho_L} + \rho_G\ \frac{\text{Re}_G\ \mu_G}{\rho_G}\right)}\ \mu_G\ ;$$

```
Simplify[μmix]
```

$$\frac{\text{Re}_G\ \mu_G^2 + \text{Re}_L\ \mu_L^2}{\text{Re}_G\ \mu_G + \text{Re}_L\ \mu_L}$$

Now substitute the averages into the expression for the dispersed pressure drop.

```
disptemp = Simplify[dpdx_disp /. {μmix -> μmix,
   Ūmix -> Umix}]
```

$$-\frac{12\,(\mathrm{Re}_G\,\mu_G^2+\mathrm{Re}_L\,\mu_L^2)\,(\mathrm{Re}_L\,\mu_L\,\rho_G+\mathrm{Re}_G\,\mu_G\,\rho_L)}{H^3\,(\mathrm{Re}_G\,\mu_G+\mathrm{Re}_L\,\mu_L)\,\rho_G\,\rho_L}$$

- **Stratified relations**

This is the ratio that we want.

```
dispstratratio = Simplify[(disptemp / strattemp) /. Re_G -> regfunc]
```

$$-((\mu_G^2\,h^4-2\,(h^3-2\,H\,h^2+3\,H^2\,h-2\,H^3)\,\mu_G\,\mu_L\,h+(h-H)^4\,\mu_L^2)\,(h\,\mu_G\,((h-4\,H)\,(h-H)^2\,\rho_G-h^3\,\rho_L)-(h-H)\,\mu_L\,((h-H)^3\,\rho_G-h^2\,(h+3\,H)\,\rho_L)))/(H^3\,(h\,\mu_G+(H-h)\,\mu_L)\,(\mu_G^2\,\rho_L\,h^4-(h-H)\,\mu_G\,\mu_L\,((h^2-5\,H\,h+4\,H^2)\,\rho_G+h\,(h+3\,H)\,\rho_L)\,h+(h-H)^4\,\mu_L^2\,\rho_G))$$

```
<< Graphics`Graphics`
```

- **Plots of the ratio**

Oil-water flow in a 1 cm channel Re_L=100. Dispersed flow has a lower pressure drop at low oil flow rates but stratified becomes lower as the oil flow rate becomes large. This is because the water *lubricates* the flow in the sense of reducing the stress near one wall. (The effect would be quantitatively larger for a pipe geometry where the water could completely surround the oil.) The reason that the pressure drop is higher for stratified at low water flow rates is that the *water*, in its stratified configuration, is taking up flow *area* that is not available to the more viscous phase.

```
sublist = {Re_L -> 100, μ_L -> .0098, μ_G -> .21,
   ρ_L -> 1, ρ_G -> .855, H -> 1}
```

$$\{\mathrm{Re}_L\to 100,\ \mu_L\to 0.0098,\ \mu_G\to 0.21,\ \rho_L\to 1,\ \rho_G\to 0.855,\ H\to 1\}$$

```
ParametricPlot[{Log[regfunc /. sublist],
  dispstratratio /. sublist}, {h, 0.05, .95}, PlotRange -> All,
 AxesLabel -> {"Log[ReG]", "dispersed/stratified ratio"}]
```



- Graphics -

Here is the plot as a function of increasing water layer thickness (based on stratified geometry). This is the same as above but the abscissa is "flipped". The total channel depth is 1.

```
Plot[(dispstratratio /. sublist),
 {h, 0.05, .95}, AxesLabel -> {"h", "dispersed/stratified ratio"}]
```



- Graphics -

Oil-water flow in a 1 cm channel Re_L =1000. The behavior is qualitatively the same for different flow conditions.

```
sublist = {ReL -> 1000, μL -> .0098, μG -> .21,
   ρL -> 1, ρG -> .855, H -> 1}
```

$\{Re_L \to 1000, \mu_L \to 0.0098, \mu_G \to 0.21, \rho_L \to 1, \rho_G \to 0.855, H \to 1\}$

```
ParametricPlot[{Log[regfunc /. sublist],
  dispstratratio /. sublist}, {h, 0.05, .95}, PlotRange -> All,
 AxesLabel -> {"Log[ReG]", "dispersed/stratified ratio"}]
```



- Graphics -

```
Plot[(dispstratratio /. sublist),
 {h, 0.05, .95}, AxesLabel -> {"h", "dispersed/stratified ratio"}]
```



- Graphics -

Here we plot the viscosity of the dispersed phase. It will be shown better below, but on the left side of the plot, the viscosity of the dispersed phase is decreasing while the pressure drop ratio is increasing. This demonstrates that the lubrication effect occurs because of the flow geometry -- less viscous fluid in region of high shear reduces the pressure drop.

```
Plot[(μmix /. ReG -> regfunc) /. sublist, {h, 0.05, .99},
 PlotRange -> All, AxesLabel -> {"h", "dispersed phase viscosity"}]
```



- Graphics -

Heavy Oil-water flow in a 1 cm channel Re_L=100. The pressure drop reduction for stratified flow is more pronounced for a more viscous oil.

```
sublist = {ReL -> 100, μL -> .0098, μG -> 5,
   ρL -> 1, ρG -> .95, H -> 1}
```

$\{Re_L \to 100, \mu_L \to 0.0098, \mu_G \to 5, \rho_L \to 1, \rho_G \to 0.95, H \to 1\}$

```
ParametricPlot[{Log[regfunc /. sublist],
  dispstratratio /. sublist}, {h, 0.01, .95}, PlotRange -> All,
 AxesLabel -> {"Log[ReG]", "dispersed/stratified ratio"}]
```



- Graphics -

```
LogLinearPlot[(dispstratratio /. sublist),
 {h, 0.01, .95}, PlotRange -> All,
 AxesLabel -> {"h", "dispersed/stratified ratio"}]
```



- Graphics -

Here we plot the viscosity of the dispersed phase. The specific lubricating region is shown on the very left of the plot where the dispersed viscosity is decreasing while the pressure drop ratio is increasing. It is also seen that there is a very great change in the mixture viscosity and a much smaller corresponding change in the pressure drop ratio.

```
LogLogPlot[(μmix / μG /. ReG -> regfunc) /. sublist,
 {h, 0.05, .99}, PlotRange -> All,
 AxesLabel -> {"h", "dispersed phase/oil viscosity"}]
```



- Graphics -

Air-water flow in a 2.54 cm channel Re_L=100. Dispersed flow always has a higher pressure drop for air-water. This differs qualitatively from the oil-water case because the large density ratio makes the mixture viscosity increase very fast as the liquid flow rate is increased. Thus there is no advantage to having a dispersed flow with the entire flow area. Note that a different mixing rule for the mixture viscosity would change this.

```
sublist = {ReL -> 100, μL -> .0098, μG -> .00018,
   ρL -> 1, ρG -> 1 / 899, H -> 2.54}
```

$$\left\{Re_L \to 100,\ \mu_L \to 0.0098,\ \mu_G \to 0.00018,\ \rho_L \to 1,\ \rho_G \to \frac{1}{899},\ H \to 2.54\right\}$$

```
ParametricPlot[
 {Log[regfunc /. sublist], dispstratratio /. sublist},
 {h, 0.001, 2.537}, PlotRange -> {0, 25},
 AxesLabel -> {"Log[ReG]", "dispersed/stratified ratio"}]
```



- Graphics -

```
Plot[(dispstratratio /. sublist),
 {h, 0.05, 2.51}, PlotRange -> All,
 AxesLabel -> {"h", "dispersed/stratified ratio"}]
```



- Graphics -

Air-water flow in a 2.54 cm channel Re_L =1000.

```
sublist = {ReL -> 1000, μL -> .0098, μG -> .00018,
   ρL -> 1, ρG -> 1 / 899, H -> 2.54}
```

$$\left\{Re_L \to 1000,\ \mu_L \to 0.0098,\ \mu_G \to 0.00018,\ \rho_L \to 1,\ \rho_G \to \frac{1}{899},\ H \to 2.54\right\}$$

```
ParametricPlot[{Log[regfunc /. sublist],
  dispstratratio /. sublist}, {h, 0.05, 2.51}, PlotRange -> All,
 AxesLabel -> {"Log[ReG]", "dispersed/stratified ratio"}]
```



- Graphics -

The maximum is not due to the specific value of the gas Reynolds number. This large maximum occurs at a depth of about 20% of H at this viscosity ratio and is probably best thought of as a geometric effect. See the dimensionless comparison below.

```
Plot[(dispstratratio /. sublist),
 {h, 0.05, 2.51}, PlotRange -> All,
 AxesLabel -> {"h", "dispersed/stratified ratio"}]
```



- Graphics -

■ **Observations**

We see a complex shape with regions where either dispersed or stratified could have a larger pressure drop. Let's explore the equation that determines this.

Again we need to get rid of the Reynolds numbers and recast the problem in terms of ratios.

```
stratdisp1 = dispstratratio /. {ReL -> UL ρL H / μL,
   ReG -> UG ρG H / μG }
```

$$-((\mu_G^2\, h^4 - 2\,(h^3 - 2\,H\,h^2 + 3\,H^2\,h - 2\,H^3)\,\mu_G\,\mu_L\,h + (h-H)^4\,\mu_L^2)\,(h\,\mu_G\,((h-4\,H)\,(h-H)^2\,\rho_G - h^3\,\rho_L) - (h-H)\,\mu_L\,((h-H)^3\,\rho_G - h^2\,(h+3\,H)\,\rho_L)))\,/\,(H^3\,(h\,\mu_G + (H-h)\,\mu_L)\,(\mu_G^2\,\rho_L\,h^4 - (h-H)\,\mu_G\,\mu_L\,((h^2 - 5\,H\,h + 4\,H^2)\,\rho_G + h\,(h+3\,H)\,\rho_L)\,h + (h-H)^4\,\mu_L^2\,\rho_G))$$

Now get it in terms of parameter ratios,

```
stratdisp2 = Simplify[stratdisp1 /. {μG -> m μL, h -> n H,
    ρG -> r ρL}]
```

$$-(((m-1)^2\,n^4+4\,(m-1)\,n^3-6\,(m-1)\,n^2+4\,(m-1)\,n+1)\,((m-1)\,(r-1)\,n^4+((4-6\,m)\,r+2)\,n^3+3\,((3\,m-2)\,r-1)\,n^2-4\,(m-1)\,r\,n-r))/(((m-1)\,n+1)\,(r\,(n-1)^4-m\,n\,((r+1)\,n^2+(3-5\,r)\,n+4\,r)\,(n-1)+m^2\,n^4))$$

Do the same for the viscosity ratio.

```
mumixratio = ((μmix / μG) /.  ReG -> regfunc) /. {ReL -> UL ρL H / μL}
```

$$\frac{1}{\mu_G}\left(\frac{H\,U_L\,\mu_L\,\rho_L}{H\,U_L\,\rho_L-\frac{(h-H)\,H\,(H-h)\,U_L\,\mu_L\,((h-H)^2\,\mu_L-h\,(h-4\,H)\,\mu_G)\,\rho_G}{h^2\,\mu_G\,(h^2\,\mu_G-(h^2+2\,H\,h-3\,H^2)\,\mu_L)}}-\frac{(h-H)\,H\,(H-h)\,U_L\,\mu_L\,((h-H)^2\,\mu_L-h\,(h-4\,H)\,\mu_G)\,\rho_G}{h^2\,(h^2\,\mu_G-(h^2+2\,H\,h-3\,H^2)\,\mu_L)\left(H\,U_L\,\rho_L-\frac{(h-H)\,H\,(H-h)\,U_L\,\mu_L\,((h-H)^2\,\mu_L-h\,(h-4\,H)\,\mu_G)\,\rho_G}{h^2\,\mu_G\,(h^2\,\mu_G-(h^2+2\,H\,h-3\,H^2)\,\mu_L)}\right)}\right)$$

Now get the viscosity ratio in terms of parameter ratios,

```
mumix2 = FullSimplify[mumixratio /. {μG -> m μL, h -> n H,
    ρG -> r ρL}]
```

$$\frac{((m\,(n-4)-n+2)\,n-1)\,r\,(n-1)^2+n^2\,(n\,(-m\,n+n+2)-3)}{-r\,(n-1)^4+m\,n\,(4\,r+n\,(n+(n-5)\,r+3))\,(n-1)-m^2\,n^4}$$

We can plot a few of these to see that the behavior is the same as complete expression.

The first plot has a density ratio of unity and the amount of more viscous fluid increases with h/H. We see that dispersed is more efficient at low h/H and that stratified has the lowest pressure drop as the more viscous fluid takes up more area.

```
LogLinearPlot[stratdisp2 /. {m -> .01, r -> 1}, {n, .01, 1},
 AxesLabel -> {"h/H", "dispersed/stratified ratio"}]
```



- Graphics -

We expand the region near the peak

```
Plot[stratdisp2 /. {m -> .01, r -> 1}, {n, .8, 1},
 AxesLabel -> {"h/H", "dispersed/stratified ratio"}]
```



- Graphics -

Now plot the viscosity ratio at the same conditions. Note that the viscosity of the dispersed phase is decreasing while the pressure drop ratio is increasing this is a clear indication of the lubrication effect! The pressure drop advantage of the stratified configuration is due to more than the specific value of the dispersed phase viscosity.

```
Plot[mumix2 /. {m -> .01, r -> 1}, {n, .8, 1},
 AxesLabel -> {"h/H", "dispersed/oil viscosity ratio"}]
```



- Graphics -

<u>The demonstration of the lubrication </u>effect is more clear if the maximum region is spread out. We can do this by simply flipping the viscosity ratio so that the bottom phase is less viscous. Now the amount of more viscous fluid <u>decreases</u> with h/H. Now the maximum occurs at low h/H and the log axis spreads this out nicely.

Plot of dispersed model/stratified model pressure drop ratio as a function of h/H, which is the depth of the less viscous phase to the channel depth. The density ratio is unity so that viscosity ratio and depth ratio are the only important parameters.

```
LogLinearPlot[stratdisp2 /. {m -> 100, r -> 1}, {n, .01, 1},
 AxesLabel -> {"h/H", "dispersed/stratified ratio"}]
```



- Graphics -

The viscosity of the dispersed phase is clearly dropping while the pressure drop ratio of dispersed/stratified is increasing. The lubrication effect, even for a channel flow where lubrication is occuring at only one wall, is thus shown!

```
LogLinearPlot[mumix2 /. {m -> 100, r -> 1}, {n, .01, 1},
 AxesLabel -> {"h/H", "dispersed/oil viscosity ratio"}]
```



- Graphics -

[BACK to dimensional comparison] that is a short way up.

[BACK to dispersed/stratified conclutions]

If the viscosity ratio is not as large the maximum moves to a larger value of the less viscous/more viscous depth ratio.

```
LogLinearPlot[stratdisp2 /. {m -> 10, r -> 1}, {n, .01, 1},
 AxesLabel -> {"h/H", "dispersed/stratified ratio"}]
```



- Graphics -

Now do air-water, dimensionless The large density ratio makes the mixture viscosity increase very fast with increases in liquid flow rate (h/H). Thus there is no region where there is a reduction in dispersed pressure drop by allowing the dispersed phase to have the entire flow area.

```
LogLinearPlot[stratdisp2 /. {m -> 1 / 55, r -> 1 / 900},
 {n, .001, 1}, AxesLabel -> {"h/H", "dispersed/stratified ratio"}]
```



- Graphics -

[BACK to Preview]
[BACK to Recap]

Here is the viscosity ratio for air-water, this should be compared to the next plot where the density ratio is set to unity. The small density ratio causes the mixture to have a much larger viscosity over the critical range where the pressure drop reduction for dispersed flow could occur.

```
LogLogPlot[mumix2 /. {m -> 1 / 55, r -> 1 / 900}, {n, .001, 1},
 AxesLabel -> {"h/H", "dispersed/gas viscosity ratio"}]
```



- Graphics -

```
LogLogPlot[mumix2 /. {m -> 1 / 55, r -> 1}, {n, .001, 1},
 AxesLabel -> {"h/H", "dispersed/gas viscosity ratio"}]
```



- Graphics -

Here we show the pressure drop ratio for the 1/55 viscosity ratio but with unity density ratio. The reduction occurs below h/H=0.5.

```
LogLinearPlot[stratdisp2 /. {m -> 1 / 55, r -> 1}, {n, .001, 1},
 AxesLabel -> {"h/H", "dispersed/stratified ratio"}]
```



- Graphics -

[BACK to Preview]
[BACK to Recap]

Conclusions for dispersed/stratified flows.

The expression for the pressure drop ratio is too complicated to find out much about. We just see from the plots that for air-water, the pressure drop is always higher for dispersed than stratified. However, for oil-water, or other cases where the density ratio is close to 1, the pressure drop for dispersed flow is lower if the more viscous fluid has a depth less than about H/2.

This can be understood by considering that for stratified flow, a thin layer of less viscous fluid "lubricates" (i. e., reduces the stress near the wall) the flow of a more viscous fluid and thus is an efficient means of transport. However, the stratified regime becomes relatively less efficient for lower ratios of the more viscous fluid because the more viscous fluid is receiving high shear and is effectively reducing the flow area available for the less viscous fluid. For these cases, a lower pressure drop is achieved with a dispersed mixture that has a lower viscosity than the high viscosity fluid, and has the entire flow area available.

We again have the overall conclusion that the degree of uncertainty in pressure drop caused by the flow regime exceeds what is desirable for good engineering design.

Stratified/Slug

Here is the ratio of dispersed flow to stratified flow

```
dpdx_strat / dpdx_slug
```

$$\frac{H^2\,\mathrm{Re}_L\,\mu_L^2\,((h-H)\,\mu_L - h\,\mu_G)}{h^2\,\mu\,\overline{U}\,((h^2+2\,H\,h-3\,H^2)\,\mu_L - h^2\,\mu_G)\,\rho_L}$$

We now need sensible values for the velocities and viscosities.

Slug relations

```
dpdx_slug
```

$$-\frac{12\,\mu\,\overline{U}}{H^2}$$

From the relation for $\mathrm{dpdx}_{\mathrm{slug}}$, we see that we need an average of the product of μ and $\overline{U}$. The way this works is that the pressure drop for each region is just $\frac{-12\,\mu_i\,U_i}{H^2}$ but the U_i is increased over its single phase velocity by the loss of flow area owing to the presence of the other phase. So in a region of L, the pressure drop is higher than if no G were present. Of course, the entire pipe is not filled with L, it is only $\frac{U_L}{(U_L+U_G)}$ full of L. So we can combine this to get the final result.

The fraction still available to fluid L should be

$$XL = 1 - \frac{U_G}{(U_L + U_G)};$$

The fraction available for fluid G is then

$$XG = 1 - \frac{U_L}{(U_L + U_G)};$$

This gives a pressure drop that becomes

$$dpslug = \frac{-12}{H^2}\left(\frac{\mu_L\,U_L}{XL}\,\frac{U_L}{(U_L+U_G)} + \frac{\mu_G\,U_G}{XG}\,\frac{U_G}{(U_L+U_G)}\right);$$

```
slugtemp = Simplify[dpslug /. {U_L -> Re_L μ_L/(H ρ_L), U_G -> Re_G μ_G/(H ρ_G)}]
```

$$-\frac{12\,(\mathrm{Re}_G\,\rho_L\,\mu_G^2+\mathrm{Re}_L\,\mu_L^2\,\rho_G)}{H^3\,\rho_G\,\rho_L}$$

This is, of course, the same as the obvious answer,

```
dpslug2 =
 FullSimplify[(-12/H^2 (μ_L U_L + μ_G U_G)) /. {U_L -> Re_L μ_L/(H ρ_L), U_G -> Re_G μ_G/(H ρ_G)}]
```

$$-\frac{12\,(\mathrm{Re}_G\,\rho_L\,\mu_G^2+\mathrm{Re}_L\,\mu_L^2\,\rho_G)}{H^3\,\rho_G\,\rho_L}$$

- **Stratified relations**

We need to make the translation from Re_L and Re_G to h and U_G and U_L.

```
regfunc = Ugexpress (H - h) ρ_G / μ_G
```

$$-\frac{(h-H)\,(H-h)\,\mathrm{Re}_L\,\mu_L^2\,((h-H)^2\,\mu_L-h\,(h-4\,H)\,\mu_G)\,\rho_G}{h^2\,\mu_G^2\,(h^2\,\mu_G-(h^2+2\,H\,h-3\,H^2)\,\mu_L)\,\rho_L}$$

```
strattemp = dpdx_strat /. {Ū_G -> Re_G μ_G / (H - h) / ρ_G, Ū_L -> Re_L μ_L / h / ρ_L}
```

$$-\frac{12\,\mathrm{Re}_L\,\mu_L^2\,((h-H)\,\mu_L-h\,\mu_G)}{h^2\,((h^2+2\,H\,h-3\,H^2)\,\mu_L-h^2\,\mu_G)\,\rho_L}$$

```
stratslugratio = Simplify[(strattemp / slugtemp) /. Re_G -> regfunc]
```

$$\frac{H^2\,(h\,\mu_G+(H-h)\,\mu_L)}{h\,(6\,h^2-9\,H\,h+4\,H^2)\,\mu_G+(-6\,h^3+9\,H\,h^2-4\,H^2\,h+H^3)\,\mu_L}$$

Plots of the ratio

Now make some plots, oil-water

```
sublist = {ReL -> 100, μL -> .0098, μG -> .21,
   ρL -> 1, ρG -> .855, H -> 1}
```

$\{\mathrm{Re}_L \to 100, \mu_L \to 0.0098, \mu_G \to 0.21, \rho_L \to 1, \rho_G \to 0.855, H \to 1\}$

We use a parametric plot to see the oil Reynolds number

```
ParametricPlot[{Log[regfunc /. sublist],
  stratslugratio /. sublist}, {h, 0.05, .95}, PlotRange -> All,
 AxesLabel -> {"Log[Reoil]", "stratified/slug ratio"}]
```



- Graphics -

Now the same one as a function of water depth. For this range, there is no significant effect of stratified compared to slug flow.

```
Plot[(stratslugratio /. sublist),
 {h, 0.001, .99}, AxesLabel -> {"h", "stratified/slug ratio"}]
```



- Graphics -

Now try air-water in 2.54 cm channel.

```
sublist = {ReL -> 1000, μL -> .0098, μG -> .0001815,
   ρL -> 1, ρG -> 1 / 890, H -> 2.54}
```

$$\left\{\text{Re}_L \to 1000,\ \mu_L \to 0.0098,\ \mu_G \to 0.0001815,\ \rho_L \to 1,\ \rho_G \to \frac{1}{890},\ H \to 2.54\right\}$$

```
ParametricPlot[{Log[regfunc /. sublist],
  stratslugratio /. sublist}, {h, 0.05, 2.5}, PlotRange -> All,
 AxesLabel -> {"Log[ReG]", "stratified/slug ratio"}]
```



- Graphics -

We check the ratio as a function of depth and again find no big difference between stratified and slug flow.

```
Plot[(stratslugratio /. sublist),
 {h, 0.05, 2.5}, AxesLabel -> {"h", "stratified/slug ratio"}]
```



- Graphics -

- **Analysis**

```
stslans1 = FullSimplify[stratslugratio /. {μG -> m μL, h -> n H}]
```

$$\frac{(m-1)\,n+1}{(m-1)\,n\,(3\,n\,(2\,n-3)+4)+1}$$

Check for the viscosity ratio being 1.

```
Simplify[stslans1 /. m -> 1]
```

1

Check for the depth ratio being 1/2

```
Simplify[stslans1 /. n -> 1 / 2]
```

1

Plot for oil-water,

```
LogLinearPlot[stslans1 /. m -> 20, {n, .001, 1}]
```



- Graphics -

Plot for air-water,

```
LogLinearPlot[stslans1 /. m -> 1 / 55, {n, .001, 1}]
```



- Graphics -

It seems that there are two interior extrema.

```
maxmin1 = FullSimplify[D[stslans1, n]]
```

$$-\frac{3\,(m-1)\,(n\,(4\,(m-1)\,n^2-3\,(m-3)\,n-6)+1)}{((m-1)\,n\,(3\,n\,(2\,n-3)+4)+1)^2}$$

```
maxmi2 = Simplify[Solve[maxmin1 == 0, n]]
```

$$\left\{\left\{n\to\frac{\frac{(m+1)^2}{\sqrt[3]{m^3-5\,m^2-5\,m+4\,\sqrt{-m\,(m^2-1)^2}+1}}+m+\sqrt[3]{m^3-5\,m^2-5\,m+4\,\sqrt{-m\,(m^2-1)^2}+1}-3}{4\,(m-1)}\right\},\right.$$

$$\left\{n\to\frac{1}{72\,(m-1)}\left(-\frac{9\,i\left(-i+\sqrt{3}\right)(m+1)^2}{\sqrt[3]{m^3-5\,m^2-5\,m+4\,\sqrt{-m\,(m^2-1)^2}+1}}+18\,(m-3)+9\,i\left(i+\sqrt{3}\right)\sqrt[3]{m^3-5\,m^2-5\,m+4\,\sqrt{-m\,(m^2-1)^2}+1}\right)\right\},$$

$$\left.\left\{n\to\frac{1}{72\,(m-1)}\left(\frac{9\,i\left(i+\sqrt{3}\right)(m+1)^2}{\sqrt[3]{m^3-5\,m^2-5\,m+4\,\sqrt{-m\,(m^2-1)^2}+1}}+18\,(m-3)-9\left(1+i\,\sqrt{3}\right)\sqrt[3]{m^3-5\,m^2-5\,m+4\,\sqrt{-m\,(m^2-1)^2}+1}\right)\right\}\right\}$$

Check the roots to see which is which.

```
maxmi2 /. m -> .10
```

$$\{\{n\to 0.244696+0.\,i\},\{n\to 1.29615-2.74129\times 10^{-17}\,i\},\{n\to 0.875818+0.\,i\}\}$$

```
maxmi2 /. m -> .01
```

$\{\{n \to 0.249447 + 0.\, i\}, \{n \to 1.06639 + 0.\, i\}, \{n \to 0.949319 + 0.\, i\}\}$

```
maxmi2 /. m -> 10.
```

$\{\{n \to 0.755304 + 4.93432 \times 10^{-17}\, i\}, \{n \to -0.296153 + 6.5791 \times 10^{-17}\, i\}, \{n \to 0.124182 - 1.31582 \times 10^{-16}\, i\}\}$

The first and third roots are the ones that are of interest. First choose the 3rd root which will give the minima.

```
stratmin1 = stslans1 /. maxmi2[[3]]
LogLinearPlot[stratmin1, {m, .01, 100}, PlotPoints -> 100,
 PlotRange -> {.2, 1}]
```



- Graphics -

We see that the minimum can be a factor of 2 or 3 for accessible parameters.

Now choose the 1st root which will give the maxima.

```
stratmax1 = stslans1 /. maxmi2[[1]]
```

```
LogLinearPlot[stratmax1, {m, .01, 100}, PlotPoints -> 100,
 PlotRange -> {1, 2}]
```



- Graphics -

We see that the maximum can be about 60% for accessible parameters. We can get the limit for m->0.

```
stratmax1 /. m -> 0
```

$$\frac{8}{5}$$

- **Conclusions for stratified verus slug**

We again see many of the features shown above. If the more viscous fluid is has a higher flow rate (depth), then stratified is a more efficient configuration and has a lower pressure drop. This point was discussed above and is globally important for the transportation of heavy oils in pipelines using the "lubricated" flow configuration where water surrounds the oil core. In contrast, if the more viscous fluid has a lower flow rate then the alternating flow takes better advantage of lower Δp for the less viscous fluid by giving it the entire flow height for fraction of the length over which it is flowing.

The magnitude of the differences are less dramatic for this example. However, the message that knowledge of the flow regime is important still should be apparent.

[BACK to Preview]
[BACK to Recap]

Stratified flow anomaly(?)

Here we find out something interesting about stratified flow.

Choose one of our favorite air water cases,

```
sublist = {ReL -> 1000, μL -> .0098, μG -> .0001815,
   ρL -> 1, ρG -> 1 / 890, H -> 2.54}
```

$$\left\{\text{Re}_L \to 1000,\ \mu_L \to 0.0098,\ \mu_G \to 0.0001815,\ \rho_L \to 1,\ \rho_G \to \frac{1}{890},\ H \to 2.54\right\}$$

We plot the stratified pressure drop versus the gas Reynolds number. It seems that there is some peculiar behavior at small Re_G. As Re_G is decreased, the pressure drop goes up. Why could this be?

```
ParametricPlot[
 {Log[regfunc /. sublist], Log[-strattemp /. sublist]}, {h, .1, 2.5},
 PlotRange -> All, AxesLabel -> {"Log[ReG]", "Log[-pressure drop]"}]
```



- Graphics -

We look a little closer, and see the definite minimum in pressure drop.

```
ParametricPlot[{Log[regfunc /. sublist],
  Log[-strattemp] /. sublist}, {h, 2.0, 2.53}, PlotRange -> All,
 AxesLabel -> {"Log[ReG]", "Log[-pressure drop]"}]
```



- Graphics -

[BACK to Preview]
[BACK to Recap]

We can't help but analyze this one. We expect it to be related to the loss of lubrication of the liquid by the gas.

```
strattemp
```

$$-\frac{12\,\mathrm{Re}_L\,\mu_L^2\,((h-H)\,\mu_L-h\,\mu_G)}{h^2\,((h^2+2\,H\,h-3\,H^2)\,\mu_L-h^2\,\mu_G)\,\rho_L}$$

```
stratextra1 = strattemp /. {ReL -> UL ρL H / μL,
    ReG -> UG ρG H / μG}
```

$$-\frac{12\,H\,U_L\,\mu_L\,((h-H)\,\mu_L-h\,\mu_G)}{h^2\,((h^2+2\,H\,h-3\,H^2)\,\mu_L-h^2\,\mu_G)}$$

Now get it in terms of parameter ratios,

```
stratextra2 = Simplify[stratextra1 /. {μG -> m μL, h -> n H,
    ρG -> r ρL}]
```

$$-\frac{12\,((m-1)\,n+1)\,U_L\,\mu_L}{H^2\,n^2\,((m-1)\,n^2-2\,n+3)}$$

We need to take the ratio of it to the single phase liquid flow

```
stratextra3 = stratextra2 / 12 / UL / μL H^2
```

$$-\frac{(m-1)\,n+1}{n^2\,((m-1)\,n^2-2\,n+3)}$$

```
LogLogPlot[-stratextra3 /. m -> .1, {n, .01, 1},
AxesLabel -> {"h/h", "stratifed/single phase ratio"}]
```



- Graphics -

We see the minimum. Adding a small amount of air to a water flow will *reduce* the pressure drop. Now take the derivative.

```
finalderiv = FullSimplify[D[stratextra3, n]]
```

$$\frac{3\,n\,((m-1)^2\,n^2+m-3)+6}{n^3\,(n\,((m-1)\,n-2)+3)^2}$$

Solve for the roots,

```
finalmin = Solve[finalderiv == 0, n]
```

$$\left\{\left\{n \to \frac{\sqrt[3]{6\sqrt{3}\,(m-1)^3\sqrt{m}\,\sqrt{m^2+18m-27}-54\,(m^2-2m+1)^2}}{3\sqrt[3]{2}\,(m^2-2m+1)} - \frac{\sqrt[3]{2}\,(m-3)}{\sqrt[3]{6\sqrt{3}\,(m-1)^3\sqrt{m}\,\sqrt{m^2+18m-27}-54\,(m^2-2m+1)^2}}\right\},\right.$$

$$\left\{n \to \frac{\left(1+i\sqrt{3}\right)(m-3)}{2^{2/3}\sqrt[3]{6\sqrt{3}\,(m-1)^3\sqrt{m}\,\sqrt{m^2+18m-27}-54\,(m^2-2m+1)^2}} - \frac{\left(1-i\sqrt{3}\right)\sqrt[3]{6\sqrt{3}\,(m-1)^3\sqrt{m}\,\sqrt{m^2+18m-27}-54\,(m^2-2m+1)^2}}{6\sqrt[3]{2}\,(m^2-2m+1)}\right\},$$

$$\left.\left\{n \to \frac{\left(1-i\sqrt{3}\right)(m-3)}{2^{2/3}\sqrt[3]{6\sqrt{3}\,(m-1)^3\sqrt{m}\,\sqrt{m^2+18m-27}-54\,(m^2-2m+1)^2}} - \frac{\left(1+i\sqrt{3}\right)\sqrt[3]{6\sqrt{3}\,(m-1)^3\sqrt{m}\,\sqrt{m^2+18m-27}-54\,(m^2-2m+1)^2}}{6\sqrt[3]{2}\,(m^2-2m+1)}\right\}\right\}$$

Check and find that the 2nd root gives reasonable values,

```
finalmin /. m -> .2
```

$\{\{n \to 1.52511 - 5.55112\times 10^{-16}\, i\}, \{n \to 0.859331 + 2.22045\times 10^{-16}\, i\}, \{n \to -2.38444 - 1.11022\times 10^{-16}\, i\}\}$

```
finalmin /. m -> .1
```

$\{\{n \to 1.2919 - 1.11022\times 10^{-16}\, i\}, \{n \to 0.879992 - 2.22045\times 10^{-16}\, i\}, \{n \to -2.17189 + 0.\, i\}\}$

```
finalmin /. m -> .01
```

$\{\{n \to 1.06626 + 1.11022\times 10^{-16}\, i\}, \{n \to 0.949447 - 2.22045\times 10^{-16}\, i\}, \{n \to -2.0157 - 5.55112\times 10^{-17}\, i\}\}$

Substitute in the 2nd root,

The location of the minimum with viscosity ratio is also interesting. It is easy to miss because it moves away from the wall only for an intemediate range.

```
Plot[n /. finalmin[[2]], {m, .001, .98},
 AxesLabel -> {"viscosity ratio", "minimum location, n"}]
```



- Graphics -

Recap of Major points

(same as the preview above)

[BACK to Preview]

We have shown, using simple models for flow regimes, stratified, slug and dispersed, that

1. The *qualitative* as well as the *quantitative* behavior of multiphase flows will change as the ratios of flow rates and physical properties change.

2. The pressure drop predictions differ substantially with flow configuration. The pressure drop for dispersed flow was predicted to be a factor of 35 higher than for slug flow in one case and a factor of 20 greater than stratified flow for another case. This key result is true for process flows and makes correct prediction of the flow regime crucial to successful design of multifluid systems. Most engineering designs cannot stand an uncertainty of a factor of 2 in the main design variable, let alone 30.

3. Stratified flow is the most efficient configuration, of the three tested here (compare stratified/slug, dispersed/stratified), for fluid transport when the more viscous fluid has a higher flow rate. This is due to the lubricating effect of the less viscous fluid that reduces shear in the more viscous fluid. This is the basis of lubricated pipeline transport of heavy oil (See D. D. Joseph and Y. Y. Renardy, *Fundamentals of Two-Fluid Dynamics*, Springer-Verlag, 1993, Vol. 2.) If the more viscous fluid is present in lessor amounts the advantage is lost because it is subjected to high shear and acts to reduce the available flow area for the less viscous fluid.

4. The loss of lubricating effect of a less viscous fluid in stratified flow can cause a region where *decreasing* the flowrate of the less viscous fluid, *increases* the pressure drop (click for specifics about *retrograde* pressure drop) -- contrary to physical intuation gained from most other flow situations.

5. The specific conclusions for dispersed/slug, dispersed/stratifed and stratified/slug can be accessed directly.

6. The reason for the differences in the pressure drop with configuration for the examples in this notebook is that the dissipation is altered. Differences in dissipation arise primarily when fluids of different viscosities are located in regions of different stress. We also find that changing the effective flow area (i.e., by having a stratified region of more viscous fluid) for the fast moving fluid changes the dissipation significantly. These general observations should hold for either laminar flow (as shown here) or turbulent flow. However, if the primary contributions to pressure drop are from fluid acceleration, or gravity, then the pressure drop differences caused by the flow regime could be less than shown here. Examples are unsteady or transient flows, developing flows or vertical flows.

Suggestions for future study

1. This notebook uses laminar flow as a basis for the calculations, it would be interesting to redo the examples using turbulent flow pressure drop relations.

2. We did not refer to the "classic" pressure drop predictions of Lockhart and Martinelli (1949) (see page 25 in the Ginoux book, or most other books on two-phase flow). It would be interesting to compare their relation with the results of this notebook, or even to develop a model analogous to the L&M one, based on the results here.

3. Probably the next major topic for study in multifluid flow, after this introduction to flow regimes, is a basic introduction to compressible flow (per suggestion of Jim Tilton from Dupont.). While I hope to have a notebook on this at sometime in the future, you can probably learn about this topic quite well using a number of good general fluid mechanics texts.

4. A number of other *Mathematica* notebooks dealing with various aspects of linear stability of multifluid flows are available at http://www.nd.edu/~mjm/.